# Understanding the computation of IntensityCentroid?

If I have an image img, and I ask ComponentMeasurements to return "IntensityValues" for the image:

m=MorphologicalComponents[Binarize[img]];
ivalues=ComponentMeasurements[{m,img},"IntensityData"];


How can I replicate the output of ComponentMeasurements[{m,img},"IntensityCentroid"]?

The challenge I'm having is understanding how to interpret a calculated intensity centroid from "IntensityValues" in the context of larger image with the overlayed morphological component? Is it flipped or rotated? It would be fantastic to see an example going from the raw "IntensityValues" data to the output of ComponentMeasurements[{m,img},"IntensityCentroid"] for the same morphological component.

-
Check out ImageMeasurements. I imagine it uses the same method as that under the hood to compute the intensity centroid or intensity values for each component. To replicate the result by applying ImageMeasurements on each component, this might be of help: mathematica.stackexchange.com/questions/28826/… –  Pickett Aug 14 '13 at 7:29
@Anon I was hoping for someone to show me how to manually compute the measurement, since I was having some trouble getting it right. –  Ynv Aug 14 '13 at 7:48

Something like this:

img     = ExampleData[{"TestImage", "Lena"}];
m       = MorphologicalComponents[Binarize[img]];
ivalues = ComponentMeasurements[{m, img}, "IntensityData"];

cents = ({#[[2]] - .5, ImageDimensions[img][[2]] - #[[1]] + .5} & /@
(Total[Position[m, #] (# /. ivalues)]/Total[(# /. ivalues)] & /@  Range[Max[m]]));


Testing:

Chop[cents - ComponentMeasurements[{m, img}, "IntensityCentroid"][[All, 2]]]


{{0,0}, {0,0}, ... {0,0}}

-
+1, but it should be ImageDimensions[img][[2]] –  Simon Woods Aug 14 '13 at 13:26
@SimonWoods You're right of course. The image coordinates are always a source of bugs. I shouldn't test with square images! Thanks. –  belisarius Aug 14 '13 at 16:20